

Bu programda DC motorun devir sayısı hesabı motor diski ile birlikte çalışan optokuplör yardımıyla bulunuyor. Motordan geribildirim sinyali optik alıcı ve verici arasına yerleştirilmiş diskin çıkışından alınmaktadır. (DSC-PORTB 4.bit)



Motorun geri beslemesi motorla beraber optik alıcı ve vericisi arasında dönen bir disk tarafından sağlanmaktadır. Diskin üzerindeki iki delik vasıtasıyla bir turda optik alıcı iki kere darbe üretmektedir. Bu darbe ile DSC çıkışında TTL dalga biçimi üretilmektedir. DSC çıkışı kullanılarak bilgisayar motorun dönme hızını ölçmekte ve değiştirebilmektedir.

Devir sayısı hesaplanırken temel mantık diskin 1 devri için geçen zamanı hesaplamak ve orantı kurarak 1 dakikadaki devir sayısını bulmaktır. Disk üzerinde 2 boşluk bulunduğundan DSC çıkışı dalga şeklinin yarı periyodunun süresini hesaplamamız yeterlidir. Daha sonra bu süre kullanılarak rpm değerini hesaplarız.

1/2 devir için geçen süreyi hesaplayan program parçası ve çevrim süreleri:

```
TEST2:  INC BX      (2)
        IN AL,UPORT1(5)
        MOV CH,FF  (2)
DELAY2:  LOOP DELAY2 (8/4)
        TEST AL,10H (3)
        JNZ TEST2  (7/3)
```

```
TEST3:  INC DX      (2)
        IN AL,UPORT1(5)
        MOV CH,FF  (2)
DELAY3:  LOOP DELAY3 (8/4)
        TEST AL,10H (3)
        JZ  TEST3   (7/3)
```



$$\{[2+5+2+(CH-1).8+4+3+7].BX+3+[2+5+2+(CH-1).8+4+3+7].DX+3\}.200ns = \frac{1 \text{ devir süresi}}{2}$$

$$\{[2+5+2+(CH-1).8+4+3+7].BX+3+[2+2+5+(CH-1).8+4+3+7].DX+3\}.400ns = 1 \text{ devir süresi}$$

1 devir süresi = a sn ise rpm = 60 sn/a sn


$$rpm = \{60sn / [(BX+DX).2063+6].400ns\}$$

hesaplamada ihmal edebiliriz.

$$rpm = \{[1,5 \cdot 10^8] / [(BX+DX).2063]\}$$

$$rpm = [72709 / (BX+DX)]$$

73171<sub>10</sub> = [0001 1C05]<sub>16</sub> (Bölme işleminde DX ve AX saklayıcılarına atanacak değerler)



Programda BX saklayıcısı tarafından TTL çıkış işaretinin  $\frac{1}{2}$  devri ifade eden bölümünün lojik"1" seviyesinde kaldığı süre karşılığında bir sayı değeri tutulur.DX saklayıcısında ise işaretin lojik"0" seviyesinde kaldığı süreyi ifade eden bir sayı değeri bulunacaktır.Bu süreler toplanarak toplam süre bulunur ve hex değer BCD ASCII değere çevrilip display e yazılır.Programda motor devri "+" ve "-" tuşlarına basılarak artırılmakta veya azaltılmaktadır.Motor durduğunda ise motorun durduğu TEST1 bölümünde sürekli olarak DSC çıkışından aynı veri değeri geldiğinde anlaşılmaktadır.Bu sürenin limit değeri 1s olacak şekilde CX saklayıcısı ve DS:1000H adresindeki değerler ile belirlenmiştir.Bu süre dolduğunda yani motorun durduğu algılandığında ekrana devir sayısı olarak sıfır yazacak olan veri girilip ekrana yazma program parçasına dallanılır ve motorun durduğu ve devir sayısının (rpm) 0 olduğu display den gözlenir.

ORG 0200H

UMODEREG EQU 86H  
 UPORTCTL1 EQU 88H  
 UPORT1 EQU 90H  
 UPORT2 EQU 92H

MOV AL,01H ;Port  
 OUT UPORTCTL1,AL ;Koşullamaları  
 MOV AL,03H ;Yapılır  
 OUT UMODEREG,AL ;

MOV AL,4DH  
 MOV AH,0CH  
 INT 28H  
 MOV AL,4DH  
 MOV AH,0CH  
 INT 28H  
 MOV AL,4FH  
 MOV AH,0CH  
 INT 28H  
 MOV AL,54H  
 MOV AH,0CH  
 INT 28H  
 MOV AL,4FH  
 MOV AH,0CH  
 INT 28H  
 MOV AL,52H  
 MOV AH,0CH  
 INT 28H  
 MOV AL,00H  
 MOV AH,0CH  
 INT 28H  
 MOV AL,00H  
 MOV AH,0CH  
 INT 28H  
 MOV AH,10H  
 MOV BX,03E8H  
 INT 28H  
 MOV AL,45H

} Display'e "MOTOR"  
 yazar ve 1s boyunca  
 bekletir.

BASLA: OUT UPORT2,AL ;AL deki veriyi DAC a bas  
 MOV BP,AX ;AX i BP ye kopyala (AL deki veriyi saklamak için)  
 TUS: MOV AH,0EH ;Tuş takımına bak.  
 INT 28 ;(AL de tuş kodu var)  
 CMP AL,0FFH ;Tuşa basıldımı kontrol et.  
 JNZ OKU ;Tuşa basıldıysa "OKU" etiketine dallan.  
 MOV DX,0H ;Veri süresini tutacak  
 MOV BX,0H ;sayaçları sıfırla.  
 MOV DS:1000H,0005H ;1000H adresine "5H" verisini ata.  
 TEKRAR: MOV CX,0FFFFH ;CX sayacını 0,2 sn bekleme süresine göre ata.  
 TEST1: IN AL,UPORT1 ;PORT1'i oku.  
 DEC CX ;sayacı "1" azalt.  
 JZ TEST4 ;sayaç değeri sıfırlandıysa "TEST4"etiketine dallan  
 TEST AL,10H ;DSC çıkışının "1" olup olmadığını test et.  
 JZ TEST1 ;Değilse TEST1'e dallan.

\*\*\*\*\*Bu kısımda motor geribildirimi için kullanılan sensör düzeneğinin ürettiği TTL dalganın yarım periyodunun toplam süresini ölçeceğiz.\*\*\*\*\*

TEST2: INC BX ;TTL dalga "1" süresi sayacını 1 arttır.  
 MOV CX,0080H ;Belirli bir süre gecikme oluştur.  
 DELAY1: LOOP DELAY1 ;(Sayaç değerini gereğinden fazla büyütmemek için)  
 IN AL,UPORT1 ;PORT1'i oku  
 TEST AL,10H ;DSC çıkışının 1 olup olmadığını test et.

```

JNZ TEST2 ;Hala "1" se TEST2'ye dallan.
TEST3: INC DX ;TTL dalga "0" süresi sayacını 1 arttır.
MOV CX,0080H ;Belirli bir süre gecikme oluştur.
DELAY2: LOOP DELAY2 ;(Sayaç değerini gereğinden fazla büyütmemek için)
IN AL,UPORT1 ;PORT1'i oku.
TEST AL,10H ;DSC çıkışının "1" olup olmadığını test et.
JZ TEST3 ;Hala "0" sa TEST3'e dallan.
ADD BX,DX ;Veri süresini tutan sayaçları topla.
MOV DX,0001H ;60s deki toplam devri bulabilmek için formüldeki
MOV AX,1DD3H ; verileri uygun saklayıcılara ata.
DIV BX ;Bölme işlemini yap.
MOV BX,AX ;Sonucu BX e kopyala

```

;\*\*\*\*\*Bu kısımda hesaplanan ve devir sayısının hex karşılığı olan sayı değerini display e basmak için veriyi BCD ASCII kodlara çevireceğiz.\*\*\*\*\*

```

CEVIR: MOV DI,0300H ;BCD ASCII kodların saklanacağı başlangıç adresini ata.
MOV CX,0006H ;Kaç digit gösterileceğini belirle.
MOV AH,09H ;BX teki hex değeri ASCII
INT 28H ;BCD değerlere çevir.
MOV AH,12H ;Display'i
INT 28H ;temizle
MOV AL,44H ;Display'e
MOV AH,0CH ;"D"
INT 28H ;yaz
MOV AL,45H ;Display'e
MOV AH,0CH ;"E"
INT 28H ;yaz
SUB DI,6 ;DI'yı 6 azalt
YAZ: MOV AL,[DI] ;DI ile adreslenen adresteki BCD ASCII kodu AL'ye al.
MOV AH,0CH ;Sayı değerini
INT 28H ;display e yaz.
INC DI ;DI yı artır
DEC CX ;CX i bütün sayı değerleri
JNZ YAZ ;display e yazılana kadar döngüyü sürdür.
MOV AH,10H ;Display e yazılan sayıları
MOV BX,10H ;BX kadar ms değerinde
INT 28H ;display de tut.
MOV AX,BP ;BP yi AX e kopyala(DAC a basılan veriyi AL ye al)
JMP BASLA ;"BASLA" etiketine dallan.
OKU: CMP AL,02BH ;"+" tuşu mu test et.
JZ ARTTIR1 ;"+" tuşu ise arttırıl e dallan.
CMP AL,02DH ;"- " tuşu mu test et.
JZ AZALT1 ;"- " tuşu ise azaltıl e dallan.
MOV AX,BP ;DAC a basılan veriyi AL ye al.
JMP BASLA ;"BASLA" etiketine dallan.
TEST4: DEC DS:1000H ;1000H adresindeki veriyi 1 azalt.
JZ CIKIS0 ;sonuç "0" sa "cıkıs0" a dallan.
JMP TEKRAR ;TEKRAR etiketine dallan.
CIKIS0: MOV BX,0000H ;Motorun durduğu algılandığı için display e
JMP CEVIR ; sıfır yazacak veriyi BX e yükle ve CEVIR e git.

ARTTIR1: MOV AX,BP ;DAC a basılan
INC AL ;veriyi 1 arttır ve
JMP BASLA ;BASLA'ya dön.
AZALT1: MOV AX,BP ;DAC a basılan
DEC AL ;veriyi 1 azalt ve
JMP BASLA ;BASLA'ya dön.

```

